A CONTINUACION ESTA UN PROGRAMA QUE PUDE BAJAR DE UN SITIO DE INTERNET AL CUAL LE HE HECHO ALGUNAS MODIFICIONES PARA LA FX-880P (EL ORIGINAL HIBA DIRIGIDO A LA 850 PERO NO VEO LA DIFERENCIA).

DEL 1 AL 16 ES TOTALMENTE MI CREACION SOLO HABRA QUE SUSTITUIR EN  DONDE DICE CLAVE(#1) LO QUE TU QUIERAS PONER Y DESPUES COLOCAR EL NUMERO DE LETRAS DE ESTA PARA VERIFICACION DEL PASSWORD(#8)

AGRADESCO AL CREADOR ORIGINAL POR ESTE ESTUPENDO PROGRAMA Y AQUÍ ESTA SU E-MAIL:

fcadela@disc.ua.es
ESTE ES EL MIO:

pikasso_j@antionline.org
CUALQUIER INQUIETUD O SUGERENCIA ENCANTADOS RECIBIREMOS SUS E-MAIL.

                                                     AGENDA

1 DATA"clave",""

2 CLEAR:CLS:PRINT"PASSWORD:";

3 F=1:GOSUB 15

4 X$=INKEY$:IF X$<CHR$(13) OR X$>CHR$(122) THEN 4

5 D$=D$+X$

6 IF X$=CHR$(13) THEN 8 ELSE 7

7 LOCATE10+N,0:PRINT"*";:N=N+1:GOTO 3

8 P$=LEFT$(D$,el numero de letras de la clave)

9 RESTORE

10 READ Y$

11 IF Y$=P$ THEN 13 ELSE 14

13 LOCATE8,1:PRINT"Acceso concedido";:BEEP1:F=5:GOSUB 15:CLS:GOTO 18

14 CLS:LOCATE10,1:PRINT"Acceso negado";:LOCATE14,0:PRINT"Adios";:F=5:GOSUB 15:

CLS:END

15 FOR J=0 TO 100*F:NEXT J

16 RETURN

18 '- AGENDA

19 CLEAR

20 ON ERROR GOTO 800

30 DIM cb$(3):cb$(1)="Telef":cb$(2)="E-mail":cb$(3)="Otros"

35 DIM pt$(2):pt$(1)="Sil":pt$(2)="Son":p=2

40 f$="$":desc$="COM0:6,N,8,1,N,N,N,B,N"

45 CLS:PRINT "Agenda 1.Bus 2.Lis 3.Nue 4.Bor";CHR$(13);"en MEMO 5.Env 6.Rec 7.";pt$(p);"8.Sal";

50 mopc=8:GOSUB 900

55 IF opc=8 THEN 80

60 ON opc GOSUB 100,400,300,600,1150,1600,1700

70 GOTO 45

80 '- FIN

85 CLS:CLEAR:END

100 '- BUSCAR

110 RESTORE#

120 bdeof=0:n=0

130 GOSUB 1100

140 CLS:INPUT "Nombre: ",c$

145 IF LEN(c$)=0 THEN RETURN

150 c$=cb$(opc)+c$

160 CLS: PRINT "...";

165 RESTORE# c$

170 IF bdeof THEN 280

180 READ# a$

185 n=n+1

190 GOSUB 1200

260 mopc=3:GOSUB 900

264 IF opc=1 THEN 160

266 IF opc=2 THEN CALC$=dt$:GOTO 260

267 RETURN

280 GOSUB 1060

285 n=0

290 RETURN

300 '- NUEVO

310 RESTORE#

320 GOSUB 1100

330 CLS:INPUT "Nombre: ",c$

335 IF LEN(c$)=0 THEN RETURN

340 CLS:INPUT "Dato: ",a$

350 c$=cb$(opc)+c$+f$+a$

360 RESTORE# c$

370 WRITE# c$

375 IF p=2 THEN BEEP

380 RETURN

400 '- LISTADO

410 RESTORE#

415 bdeof=0

420 GOSUB 1100

425 c$=cb$(opc)

430 CLS:PRINT "...";

435 RESTORE# c$

440 IF bdeof THEN 550

450 READ# a$

460 GOSUB 1200

540 mopc=3:GOSUB 900

544 IF opc=1 THEN 430

546 IF opc=2 THEN CALC$=dt$:GOTO 540

547 RETURN

550 GOSUB 1060

560 RETURN

600 '- BORRAR

610 GOSUB 100

620 IF n=0 THEN RETURN

630 CLS:PRINT "1.No borrar y volver al menu";CHR$(13);"2.Borrar ultimo dato";

640 mopc=2:GOSUB 900

660 IF opc=1 THEN RETURN

670 bdeof=0:RESTORE#

680 FOR i=2 TO n

690 RESTORE# c$

700 IF bdeof THEN RETURN

710 READ# a$

720 NEXT i

730 RESTORE# c$

740 WRITE# 

745 IF p=2 THEN BEEP

750 RETURN

800 '- ERROR

810 l=ERR

820 IF l=22 THEN 850

830 PRINT "Error";l;"en la linea";ERL;

835 IF p=2 THEN BEEP

840 END

850 bdeof=-1

860 ON ERROR GOTO 800

870 RESUME NEXT

900 '- LEER OPCION

905 LOCATE 31,1

910 a$=INPUT$(1)

915 opc=ASC(a$)-48

920 IF opc<1 OR opc>mopc THEN 910

925 IF p=2 THEN BEEP1

930 RETURN

1000 '- f$ EN a$

1005 l=LEN(a$)

1010 FOR i=1 TO l

1020 IF MID(a$,i,1)=f$ THEN 1050

1030 NEXT i

1040 i=0

1050 RETURN

1060 '- NO MAS

1064 IF p=2 THEN BEEP

1070 CLS:PRINT "No hay mas entradas      1.Menu";

1080 mopc=1:GOSUB 900

1090 RETURN

1100 '- MENU TIPO

1110 CLS:PRINT "Elige tipo de dato:";CHR$(13);"1.Telefono 2.E.mail 3.Otro";

1130 mopc=3:GOSUB 900

1140 RETURN

1150 '- OPC ENVIO

1160 CLS:PRINT "1.Enviar listado con formato";CHR$(13);"2.Enviar datos";

1170 mopc=2:GOSUB 900

1180 ON opc GOSUB 1300,1500

1190 RETURN

1200 '- VER REG

1210 a$=RIGHT$(a$,LEN(a$)-1)

1215 GOSUB 1000

1220 b$="?"

1225 IF i=0 THEN 1250

1230 b$=RIGHT$(a$,LEN(a$)-i):dt$=b$

1235 IF LEN(b$)>25 THEN b$=LEFT$(b$,25)

1240 i=i-1: IF i>24 THEN i=24

1245 a$=LEFT$(a$,i)

1250 CLS:PRINT a$;

1255 LOCATE 30,1:PRINT b$;

1260 LOCATE 17,1:PRINT "1.S 2.MIn 3.F";

1268 IF p=2 THEN BEEP

1270 RETURN

1280 '- INI COM

1284 CLS:PRINT "Listo para envio a 4800bps 8bit";CHR$(13);"No Paridad, 1 Bit Stop, XOnXOff"

1286 CLS:PRINT "Enviando datos...";

1290 OPEN desc$ AS #1

1294 RETURN

1300 '- ENVIAR1

1310 GOSUB 1280

1314 PRINT #1,"DATOS DE lA AGENDA";CHR$(13);CHR$(10);

1316 FOR opc=1 TO 3

1318 IF opc=1 THEN a$="Telefonos" ELSE IF opc=2 THEN a$="E-mails" ELSE a$="Otros"

1320 PRINT #1,"- ";a$;" -"

1322 a$=CHR$(5)+a$+" Registro"

1324 r=LEN(a$)

1326 LOCATE 0,1:PRINT a$;

1328 RESTORE#

1330 bdeof=0:n=0

1334 RESTORE# cb$(opc)

1340 IF bdeof THEN 1420

1342 n=n+1:LOCATE r,1:PRINT n;

1346 READ# a$

1360 a$=RIGHT$(a$,LEN(a$))

1365 GOSUB 1000

1370 b$="?"

1375 IF i=0 THEN 1390

1380 b$=RIGHT$(a$,LEN(a$)-i)

1384 a$=LEFT$(a$,i)

1390 PRINT #1,n;a$;" : ";b$

1400 GOTO 1334

1402 CLOSE

1420 NEXT opc

1430 PRINT #1,"- FIN -":PRINT #1

1444 IF p=2 THEN BEEP

1450 RETURN

1500 '- ENVIAR2

1505 GOSUB 1280

1510 PRINT CHR$(13);"Registro";

1515 RESTORE#

1520 bdeof=0:n=0

1525 READ# a$

1530 IF bdeof THEN 1560

1532 n=n+1:LOCATE 8,1:PRINT n;

1534 b$=LEFT$(a$,1)

1536 a$=RIGHT$(a$,LEN(a$))

1538 c$=b$+f$+a$

1540 PRINT #1,c$

1555 GOTO 1525

1560 PRINT #1

1562 CLOSE

1564 IF p=2 THEN BEEP

1566 RETURN

1600 '- RECIBIR

1602 CLS:PRINT "1.No recibir y volver al menu";CHR$(13);"2.Borrar datos y recibir";

1604 mopc=2:GOSUB 900

1606 IF opc=1 THEN RETURN

1610 GOSUB 1800

1612 CLS:PRINT "Listo para recibir 4800bps 8bit";CHR$(13);"No Paridad, 1 Bit Stop, XOnXOff"

1614 CLS:PRINT "Recibiendo datos...";CHR$(13);"Registro";

1620 n=0:OPEN desc$ AS #1

1625 INPUT #1,a$

1630 IF LEN(a$)<3 THEN 1680

1634 n=n+1:LOCATE 8,1:PRINT n;

1636 b$=LEFT$(a$,1)

1638 a$=RIGHT$(a$,LEN(a$)-2)

1640 c$=b$+a$

1644 RESTORE# c$

1660 WRITE# c$

1665 GOTO 1625

1680 CLOSE

1685 IF p=2 THEN BEEP

1690 RETURN

1700 '- PITO

1710 p=p+1:IF p=3 THEN p=1

1720 IF p=2 THEN BEEP

1730 RETURN

1800 '- BORRAR

1810 bdeof=0

1815 RESTORE#:WRITE#

1820 RESTORE#:READ# a$

1825 IF bdeof THEN 1845

1830 GOTO 1815

1845 RETURN

